**graph twoway fpfit — Twoway fractional-polynomial prediction plots**

### Description

twoway fpfit calculates the prediction for yvar from estimation of a fractional polynomial of xvar and plots the resulting curve.

### Quick start

Fractional-polynomial prediction plot of predicted y on x using regress

```
twoway fpfit y x
```

As above, but estimate a fractional polynomial of degree 4

```
twoway fpfit y x, estopts(degree(4))
```

Overlaid on a scatterplot of the observed values

```
twoway scatter y x || fpfit y x
```

As above, with lines for fractional polynomials of degrees 2 (default) and 4

```
twoway scatter y x || fpfit y x || fpfit y x, estopts(degree(4))
```

A separate graph area for each level of catvar

```
twoway scatter y x || fpfit y x, by(catvar)
```

As above, but with an additional graph area for all levels of catvar combined

```
twoway scatter y x || fpfit y x, by(catvar, total)
```

Overlay lines and data points for catvar==0 and catvar==1 in a single graph area

```
twoway scatter y x if catvar==0 || scatter y x if catvar==1 || ///
    fpfit y x if catvar==0 || fpfit y x if catvar==1
```

Specify poisson as the estimation command

```
twoway scatter y x || fpfit y x, estcmd(poisson)
```

### Menu

Graphics > Twoway graph (scatter, line, etc.)
Syntax

```
twoway fpfit yvar xvar [if] [in] [weight] [ , options ]
```

<table>
<thead>
<tr>
<th>options</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>estcmd(est_cmd)</td>
<td>estimation command; default is regress</td>
</tr>
<tr>
<td>estopts(est_opt)</td>
<td>specifies est_opt to estimate the fractional polynomial regression</td>
</tr>
<tr>
<td>cline_options</td>
<td>change look of predicted line</td>
</tr>
<tr>
<td>axis_choice_options</td>
<td>associate plot with alternative axis</td>
</tr>
<tr>
<td>twoway_options</td>
<td>titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.</td>
</tr>
</tbody>
</table>

*est_cmd* may be clogit, glm, intreg, logistic, logit, mlogit, nbreg, ologit, oprobit, poisson, probit, regress, rreg, stcox, stcrreg, streg, or xtgee.


Options estcmd() and estopts() are *unique*; see [G-4] Concept: repeated options.

aweights, fweights, and pweights are allowed. Weights, if specified, affect estimation but not how the weighted results are plotted. See [U] 11.1.6 weight.

*est_opt*  
<table>
<thead>
<tr>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>degree(#)</td>
</tr>
<tr>
<td>noscaling</td>
</tr>
<tr>
<td>noconstant</td>
</tr>
<tr>
<td>powers(numlist)</td>
</tr>
<tr>
<td>center(cent_list)</td>
</tr>
<tr>
<td>all</td>
</tr>
<tr>
<td>log</td>
</tr>
<tr>
<td>compare</td>
</tr>
<tr>
<td>display_options</td>
</tr>
<tr>
<td>other_est_opt</td>
</tr>
</tbody>
</table>

*cent_list* is a comma-separated list with elements varlist: {mean | # | no}, except that the first element may optionally be of the form {mean | # | no} to specify the default for all variables.

Options

*estcmd(est_cmd)* specifies the estimation command to be used; estcmd(regress) is the default.

*estopts(est_opt)* specifies the options to estimate the fractional polynomial regression from which the curve will be predicted. Available *est_opt* are

- degree(#) determines the degree of FP to be fit. The default is degree(2), that is, a model with two power terms.
nosingning suppresses scaling of $xvar_1$ and its powers.

noconstant suppresses the regression constant if this is permitted by est_cmd.

powers(numlist) is the set of FP powers from which models are to be chosen. The default is powers(-2,-1,-.5,0,.5,1,2,3) (0 means log).

center(cent_list) defines the centering for the covariates $xvar_1$, $xvar_2$, ..., $xvarlist$. The default is center(mean). A typical item in cent_list is varlist: {mean | # | no}. Items are separated by commas. The first item is special because varlist: is optional, and if omitted, the default is (re)set to the specified value (mean or # or no). For example, center(no, age:mean) sets the default to no and sets the centering for age to mean.

all includes out-of-sample observations when generating the best-fitting FP powers of $xvar_1$, $xvar_2$, etc. By default, the generated FP variables contain missing values outside the estimation sample.

log displays deviances and (for regress) residual standard deviations for each FP model fit.

cmpare reports a closed-test comparison between FP models.

display_options: cformat(%fmt), pformat(%fmt), sformat(%fmt), and nolstretch; see [R] Estimation options.

other_est_opts are options appropriate to the est_cmd; see the documentation for that est_cmd. For example, for stcox, other_est_opts may include efron or some alternate method for handling tied failures.

cline_options specify how the prediction line is rendered; see [G-3] cline_options.

axis_choice_options associate the plot with a particular y or x axis on the graph; see [G-3] axis_choice_options.

twoway_options are a set of common options supported by all twoway graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over by() groups, and change some advanced settings. See [G-3] twoway_options.

Remarks and examples stata.com

Remarks are presented under the following headings:

Typical use
Cautions
Use with by()
**Typical use**

`twoway fpfit` is nearly always used in conjunction with other `twoway` plottypes, such as

```
. use https://www.stata-press.com/data/r16/auto
(1978 Automobile Data)
. scatter mpg weight || fpfit mpg weight
```

**Cautions**

Do not use `twoway fpfit` when specifying the `axis_scale_options` `yscale(log)` or `xscale(log)` to create log scales. Typing

```
. scatter mpg weight, xscale(log) || fpfit mpg weight
```

will produce a curve that will be fit from a fractional polynomial regression of `mpg` on `weight` rather than `log(weight)`. 
Use with by()

`fpfit` may be used with by() (as can all the `twoway` plot commands):

```
.scatter mpg weight || fpfit mpg weight ||, by(foreign, total row(1))
```

![Graphs by Car type](image)

Also see

[G-2] `graph twoway fpfitci` — Twoway fractional-polynomial prediction plots with CIs

[G-2] `graph twoway line` — Twoway line plots

[G-2] `graph twoway lfit` — Twoway linear prediction plots

[G-2] `graph twoway qfit` — Twoway quadratic prediction plots

[G-2] `graph twoway mband` — Twoway median-band plots

[G-2] `graph twoway mspline` — Twoway median-spline plots